

Query Wikidata in SPARQL

34th Chaos Communication
Congress, Leipzig 2017



WIKIMEDIA
DEUTSCHLAND

Please interrupt me!

I might get carried away, so please feel free to ask question at any time :)

(At some point I'll run out of slides and we'll transition into Q&A-mode anyways.)



Interesting questions


- [How many popes had children?](#)
- [What's the largest city with a female mayor?](#)
- [Which films were set in a future which is now the past?](#)
- [Which awards were won by more women than men?](#)



First queries

The simplest way to get started with some queries is to take an example query and start editing it – even without SPARQL!











<http://query.wikidata.org>



The screenshot shows the Wikidata Query interface. At the top, there is a navigation bar with the Wikidata logo, the text "Wikidata Query", and buttons for "Examples", "Help", and "Tools". The "Examples" button is highlighted with an orange arrow. Below the navigation bar, there is a "Query Helper" panel on the left with a close button (an 'x' in a circle) and options for "+ Filter", "+ Show", and "Limit 100". The main area on the right contains a text input field with the placeholder text "1 (Input a SPARQL query or choose a query example)".

head of government (P6) start time (P580)
 country of citizenship (P27) country (P17) point in time (P585) city (Q515)
 date of death (P570) position held (P39) coordinate location (P625) place of birth (P19)
 inception (P571) female (Q6581072) **instance of (P31)** date of birth (P569) human (Q5)
 image (P18) subclass of (P279) occupation (P106) end time (P582) child (P40)
 WikiPathways ID (P2410) genre (P136) located in the administrative territorial entity (P131)

▼ Type to filter 435

- Cats   
- Goats 
- Horses 
- Wikidata items with a Wikispecies sitelink 
- Wikidata items about authors with a Wikispecies page 
- Recent Events 
- Popular eye colors 
- Even more cats, with pictures 





Query Helper

instance of

cat
hackerspace

hackerspace (Q1032372)
community-operated physical space where people with common interests, often in computers, technology, science, digital art or electronic art. can

```
1 #Cats
2 SELECT ?item ?itemLabel
3 WHERE
4 {
5   ?item wdt:P31 wd:Q146.
6   SERVICE wikibase:label { bd:serviceParam wi
7 }
8
9 (^ ignore ^)
```





Query Helper



instance of

hackerspace

+ Filter

country

Germany

image

+ Show

coordinate location

Limit

```

1 #Cats
2 SELECT ?item ?itemLabel ?image ?coordinate_lo
3   ?item wdt:P31 wd:Q1032372.
4   SERVICE wikibase:label { bd:serviceParam wi
5   ?item wdt:P17 wd:Q183.
6   OPTIONAL { ?item wdt:P18 ?image. }
7   OPTIONAL { ?item wdt:P625 ?coordinate_locat
8 }

```



RDF

- WDQS queries operate on Wikidata's data mapped to *subject – predicate – object* triples
- for example,
 c-base (Q265688) instance of (P31) hackerspace (Q1032372)
looks like
 wd:Q265688 wdt:P31 wd:Q1032372.



First queries

To check whether this statement exists:

```
ASK {  
  wd:Q265688 wdt:P31 wd:Q1032372.  
}
```



First queries

To find *all* hackerspaces:

```
SELECT ?hackerspace WHERE {  
  ?hackerspace wdt:P31 wd:Q1032372.  
}
```

“Find all possible values for ?hackerspace where the resulting statement exists.”



First queries

...all hackerspaces in Germany:

```
SELECT ?hackerspace WHERE {  
  ?hackerspace wdt:P31 wd:Q1032372.  
  ?hackerspace wdt:P17 wd:Q183.  
}
```

“where both statements exist”



Autocompletion

The Wikidata Query Service has awesome autocompletion. You don't need to memorize any Q- or P-numbers – just type `wd:hackerspace` (for items) or `wdt:country` (for properties), hit Ctrl+Space, and select the right suggestion from the autocompletion.



First queries

Back to the hackerspaces, now with labels:

```
SELECT ?hackerspace ?hackerspaceLabel WHERE {  
  ?hackerspace wdt:P31 wd:Q1032372.  
  ?hackerspace wdt:P17 wd:Q183.  
  # do not memorize ↓ – autocomplete SERVICE!  
  SERVICE wikibase:label { bd:serviceParam wikibase:language "[AUTO_LANGUAGE],en". }  
}
```



Other events at CCL

Challenge: find items with “location” “Leipzig fairgrounds”.
You can use the Query Helper or try writing SPARQL yourself.

(For now, we don't care whether they're events or anything else. Any item counts!)



Other events at CCL

Challenge: find items with “location” “Leipzig fairgrounds”.

```
SELECT ?event ?eventLabel WHERE {  
  ?event wdt:P276 wd:Q46967564.  
  
  SERVICE wikibase:label { bd:serviceParam wikibase:language "[AUTO_LANGUAGE],en". }  
}
```



Other events at CCL

Add “start time”, if known:

```
SELECT ?event ?eventLabel ?startTime WHERE {  
  ?event wdt:P276 wd:Q46967564.  
  OPTIONAL { ?event wdt:P580 ?startTime. }  
  SERVICE wikibase:label { bd:serviceParam wikibase:language "[AUTO_LANGUAGE],en". }  
}
```



Other events at CCL

Filter for instances of “event” (including subclasses):

```
SELECT ?event ?eventLabel ?startTime WHERE {  
  ?event wdt:P31/wdt:P279* wd:Q1656682;  
         wdt:P276 wd:Q46967564.  
  OPTIONAL { ?event wdt:P580 ?startTime. }  
  SERVICE wikibase:label { bd:serviceParam wikibase:language "[AUTO_LANGUAGE],en". }  
}
```



CCC on Wikidata

How many instances of “Chaos Communication Congress” exist on Wikidata? What other statements do they have?



CCC on Wikidata

For example:

```
SELECT ?congress ?congressLabel ?motto ?start ?end ?participants WHERE {  
  ?congress wdt:P31 wd:Q516804.  
  OPTIONAL { ?congress wdt:P1451 ?motto. }  
  OPTIONAL { ?congress wdt:P580 ?start. }  
  OPTIONAL { ?congress wdt:P582 ?end. }  
  OPTIONAL { ?congress wdt:P1132 ?participants. }  
  SERVICE wikibase:label { bd:serviceParam wikibase:language "[AUTO_LANGUAGE],en". }  
}  
ORDER BY ?start
```



Hackerspaces

Back to hackerspaces – how many of them are in Wikidata for each country?

First, we get a list of all hackerspaces with country that Wikidata knows about.



Hackerspaces

```
SELECT ?hackerspace ?hackerspaceLabel ?countryLabel WHERE {  
  ?hackerspace wdt:P31/wdt:P279* wd:Q1032372;  
  wdt:P17 ?country.  
  
  SERVICE wikibase:label { bd:serviceParam wikibase:language "[AUTO_LANGUAGE],en". }  
}  
ORDER BY ?countryLabel
```



Hackerspaces

Then, for each country we collapse all the results into a single one. We can no longer select the ?hackerspaces directly (there's more than one value and we only have a single result), but we can count them instead.



Hackerspaces

```
SELECT ?countryLabel (COUNT(?hackerspace) AS ?count) WHERE {  
  ?hackerspace wdt:P31/wdt:P279* wd:Q1032372;  
  wdt:P17 ?country.  
  
  SERVICE wikibase:label { bd:serviceParam wikibase:language "[AUTO_LANGUAGE],en". }  
}  
GROUP BY ?countryLabel
```



Thank you!

- me: [@WikidataFacts](https://www.wikidata.org/wiki/Wikidata:User:@WikidataFacts)
- slides: tinyurl.com/wdqs-34c3
- help: wikidata.org/wiki/Help:SPARQL
- request a query: wikidata.org/wiki/WD:RAQ

